Project
Optimal Control

Martin Hast
martin.hast@control.lth.se

July 10, 2014

1. Introduction

As the project in optimal control I have chosen to model a rocket. The model includes gravity, drag, varying air temperature, pressure and density, and, fuel consumption. The German V2-rocket[1] serves as model, for the model, and for the parameters that I couldn't find in a data sheet I have made more or less qualified guesses. The V2-rocket was chosen since it is well-known and information about it is easy to find. The control signals to be determined are the thrust magnitude, u, and the thrust angle θ.

The optimization problem is to launch the rocket and to guide it to a specific coordinate in the minimum possible time.

The optimization problem are solved using JModelica.

2. Modeling

Gravity

The gravitational force acting on the rocket is given by

$$F_g = mg \tag{1}$$

where g is the acceleration due to gravity and

$$m(t) = m_r + f(t). \tag{2}$$

m_r is the mass of the rocket excluding the fuel, $f(t)$.

Drag

The force due to air resistance (drag[2]) acting on the rocket is modeled as

$$F_d = \frac{1}{2}\rho(p, T)v^2 A C_d \tag{3}$$

where A is the rocket cross-section, v is the rocket speed. I have not been able to find the drag coefficient, C_d, for the V2-rocket but for a typical model rocket it is 0.75[3] which I used in the modeling. The air density $\rho(p, T)$ is a function of both the air pressure p and the air temperature T.

Air temperature, pressure and density

The model of air temperature and density is based on the model that can be found in the Wikipedia article on air density[4].

The temperature at y meters above sea level is given by

$$T(y) = T_0 - Ly \tag{4}$$

where T_0 is the temperature at sea level, measured in Kelvin, and L is the temperature lapse rate. This approximate model is only valid inside the troposphere.

[1] http://en.wikipedia.org/wiki/V-2_rocket
[2] http://en.wikipedia.org/wiki/Drag_(physics)
[3] http://en.wikipedia.org/wiki/Drag_coefficient
[4] http://en.wikipedia.org/wiki/Density_of_air

The temperature at sea level is assumed to be 15°C which gives $T_0 = 288.15$. The temperature lapse rate is approximately $6.5 \cdot 10^{-3}$. Since the absolute temperature can never be less than zero this model is only valid for altitudes

$$y \leq \frac{T_0}{L} \approx 44330. \tag{5}$$

The rocket can never travel under ground so $y \geq 0$.

The pressure at y meters above sea level is given by

$$p(y) = p_0 \left(1 - \frac{Ly}{T_0}\right)^{\frac{gM}{RL}} \tag{6}$$

where p_0 is the pressure at sea level, M is the molar mass of air and R is the ideal gas constant.

The density of air as functions of the pressure and temperature is given by

$$\rho(p, T) = \frac{pM}{RT} \tag{7}$$

V2-rocket - fuel consumption and thrust

The V2-rocket was 14 meters long, with a diameter of 1.65 meters and weighed 12,500 kg. It carried a 1000 kg warhead of Amatol[5] and 8720 kg of propellant. The propellant consisted of 3810 kg of an ethanol and water mix and 4,910 kg of liquid oxygen. To simplify the modeling I have assumed that the rocket only has one type of propellant and that the fuel mass f decreases according to

$$\dot{f} = -\frac{u}{c} \tag{8}$$

where u is a (normalized) control signal and c is a constant.

When fired, the V2-rocket burnt for approximately 60 s[6]. Assuming that maximum control signal is applied during 60 s and that there is no fuel left after that period, the constant c can be determined. Solving

$$\frac{df}{dt} = -\frac{1}{c}, \quad f(0) = 8720, \ f(60) = 0 \tag{9}$$

gives

$$c = \frac{3}{436}. \tag{10}$$

I assume that the force accelerating the rocket is given by

$$F_a = au \tag{11}$$

where a is a constant that I need to determine and u is the normalized control signal.

To find a the following assumptions are made

- The rocket starts at rest and accelerates with $u = 1$.
- The rocket accelerates in a straight line, perpendicular to the gravitational force

[5] `http://en.wikipedia.org/wiki/Amatol`

[6] `http://www.v2rocket.com/start/makeup/design.html`

Name	Value	Unit	Description
g	9.82	m/s^2	Acceleration due to gravity
T_0	288.15	K	Sea level temperature
L	$6.5 \cdot 10^{-3}$	K/m	Temperature lapse rate
R	8.32	J/K/mol	Ideal gas constant
M	$2.90 \cdot 10^{-2}$	kg/mol	Air molar mass
p_0	$101.325 \cdot 10^3$	Pa	Sea level atmospheric pressure
m_r	3780	kg	Rocket mass
$f(0)$	8720	kg	Initial fuel
d	1.65	m	Rocket diameter
A	2.14	m^2	Rocket cross-section
C_d	0.75	1	Drag coefficient
c	3/436	s/kg	Fuel consumption rate
a	$3.00 \cdot 10^6$	m/s^2	Thrust coefficient

Table 1 Model constants.

- The rocket travels at sea level so that the temperature, air pressure and air density is constant, $k \equiv \rho A C_D = 1.97$.

Denoting the speed by v_x, a differential equation that describes the speed is given by

$$\left(12500 - \frac{t}{c}\right)\dot{v}_x(t) = a - \frac{k}{2}v_x(t)^2 \tag{12}$$

Solving this with the initial conditions $v_x(0) = 0$ gives

$$v_x(t) = \sqrt{\frac{a}{k}}\tanh\left(c\sqrt{ak}\ln\left(\frac{12500c}{12500c - t}\right)\right). \tag{13}$$

The V2-rocket obtained a speed of 1341 m/s when the fuel was exhausted[7]. Assuming that the rocket was accelerated for 60 seconds before and that it attained it maximum speed we obtain

$$v_x(60) = 1341 \tag{14}$$

from which we can solve for a to obtain

$$a = 1.77 \cdot 10^6. \tag{15}$$

A lot of assumptions, some better than others, but still a constant that hopefully should be in the correct order of magnitude.

All model constants are summarized in Table 1 and the variables are summarized in Table 2.

[7] http://www.v2rocket.com/start/makeup/design.html

Name	Unit	Description
x	m	x-coordinate
y	m	y-coordinate, altitude
v	m/s	speed
f	kg	Fuel mass
p	Pa	Pressure
T	K	Temperature
ρ	kg/m^3	Air density
F_d	N	Drag force
F_g	N	Gravitational force
u	1	Input signal, thrust magnitude
θ	rad	Input signal, thrust angle

Table 2 Model variables.

3. Dynamical equations

The position of the rockets is expressed by the its x and y coordinates. The force equation are

$$m(t)\ddot{x}(t) = (au(t) - F_d(t)) \cdot \cos(\theta(t)) \tag{16}$$
$$m(t)\ddot{y}(t) = (au(t) - F_d(t)) \cdot \sin(\theta(t)) - F_g \tag{17}$$

The rocket position at launch is the origin of the coordinate system

$$x(0) = 0, \quad y(0) = 0 \tag{18}$$

and the initial speed is zero

$$\dot{x}(0) = 0, \quad \dot{y}(0) = 0. \tag{19}$$

The gravitational force is given by (1). the fuel consumption by (9), the drag is given by (3) together with (4), (6) and (7). Finally, the mass of the rocket is given by (2).

4. Optimization problem

As mentioned above the thrust magnitude is normalized i.e.

$$0 \leq u \leq 1. \tag{20}$$

Furthermore, the thrust angle is constrained to

$$-\frac{\pi}{2} \leq \theta \leq \frac{\pi}{2}. \tag{21}$$

This is justified by the fact that it makes no sense, in the following optimization problem, that the rocket travels "backwards". It also makes it easier for the solver to obtain a solution.

It was noticed that the optimal solutions found were very oscillatory. To limit the oscillations and obtain reasonable control signal trajectories, the inputs in the optimization problem are the derivative of u and θ respectively. Both derivatives were limited to the interval $[-0.3, 0.3]$.

The V2-rocket had an operational range of 320 km. It used gyroscopes to determine the direction during flight and a Müller-type pendulous gyroscopic accelerometer[8] to determine engine cut-off. Direction and thrust could not be changed during the flight which should limit the operational range. Since I assume that both thrust magnitude and angle can be changed during flight it is reasonable to assume that the operational range should be much larger. In the optimization problem solved the target is located ten times further away, on sea level i.e.,

$$x(t_f) = 3.2 \cdot 10^6, \quad y(t_f) = 0 \tag{22}$$

where t_f is the (free) final time.

Time optimal control

It is reasonable to assume that you would like the rocket to reach its final destination as fast as possible, thus minimizing the risk of it being shot down. The cost function to be minimized is

$$J = t_f. \tag{23}$$

JModelica fails to find a solution to the minimum time problem if not a good initial guess is provided. Therefore, a sequence of optimization problems is solved. First a simulation with constant $u = 1$ and $\theta = \frac{\pi}{10}$ is performed. The result is then used as an initial guess to solve the optimization problem with the cost function

$$\min_{u,\theta} J = t_f + 0.2 \int_0^{t_f} \dot{u}^2 + \dot{\theta}^2 dt. \tag{24}$$

This problem is first solved using 30 elements and the inputs are constant between in each element. The solution to this problem is then used to solve the same problem but this time with 50 element. Lastly, this solution is used as the initial guess to solve the original problem.

Result

The minimal time that it takes to reach the final destination is

$$t_f^* = 303\,\mathrm{s}. \tag{25}$$

The optimal trajectory can be seen in Figure 1. The altitude as function of time can be seen in Figure 2. The optimal inputs as function of time can be seen in Figure 3 and Figure 4. Figure 5 shows how the temperature and air pressure varies as function of time.

It has been anything but trivial to provide the solver with good initial guesses and according to Fredrik, I would have benefited from a better solver. Unfortunately, I did not have access to one.

[8] http://en.wikipedia.org/wiki/Muller-type_pendulous_gyroscopic_accelerometer



Figure 1 The time optimal trajectory in blue. The initial optimization with 30 elements in green and simulation result in red.



Figure 2 Optimal altitude as function of time.



Figure 3 Optimal thrust magnitude as function of time.



Figure 4 Optimal thrust angle as function of time.



Figure 5 Optimal air temperature and air pressure as function of time.

A. JModelica code

The Modelica code used to solve the problem(s) is provided below.

```
model Rocket
    constant Real Pi = Modelica.Constants.pi;
    //
    parameter Real g = 9.81;        // Gravititional constant   [m/s^2]
    parameter Real T0 = 288.15;     // Sea level temperature    [K]
    parameter Real L = 6.5e-3;      // Temperature lapse rate   [K/m]
    parameter Real R = 8.31;        // Ideal gas constant       [J/K/mol]
    parameter Real M = 0.02896;     // Molar mass           [kg/mol]
    parameter Real p0 = 101.325e3;  // Sea level atmospheric pressure [Pa]
    // V2-rocket parameters
    parameter Real rocket_mass = 3780;  // Mass excluding fuel      [kg]
    parameter Real d = 1.65;        // Diameter           [m]
    parameter Real A = (d/2)^2*Pi;  // Cross-section          [m^2]
    parameter Real Cd = 0.75;       // Drag coefficient       [1]
    parameter Real c = 3/436;       // Fuel consumption coefficient [s/kg]
    parameter Real a = 1.77e6;      // Trust constant         [m/s^2]
    // Variables
    Real m(min=rocket_mass);                        // Total mass [kg]
    Real fuel(start=8720, fixed=true, min=0);   // Fuel        [kg]

    Real y(start=0, fixed=true, min=0, max=44330);  // Altitude
[m]
    Real v_y(start=0, fixed=true);                      // Velocity in y-direction
[m/s]
    Real x(start=0, fixed=true);                        // x-position
[m]
    Real v_x(start=0, fixed=true, min=0);               // Velocity in x- direction [m/s]
    Real v_squared;                                     // Speed squared
[(m/s)^2]

    Real p(min=0);   // Pressure         [Pa]
    Real rho(min=0); // Air density      [kg/m^3]
    Real T(min=0);   // Air temperature [K]

    Real Fd;         // Drag force   [N]
    Real Fg;         // Gravitational force [N]
    // Inputs
    Real u(start=1, min=0, max=1);                    // Control magnitude [1]
    Real theta(start=Pi/10, min=-Pi/2, max=Pi/2); // Control angle
[rad]

    input Real du(min=-0.3, max=0.3);               // Derivative of u
    input Real dtheta(min=-0.3,max=0.3);            // Derivative of theta

equation
    // Mass and fuel
    m = rocket_mass + fuel;
    der(fuel) = -u/c;

    // Air temperature, pressure and density
    p = p0*(1-L*y/T0)^(g*M/(R*L));
    T = T0-L*y;
    rho = p*M/(R*T);

    // Forces acting on the rocket
    Fg = m*g;
```

```
    Fd = 0.5*rho*v_squared*A*Cd;

    // Rocket velocity and acceleration
    der(y) = v_y;
    der(v_y)*m = (a*u - Fd)*sin(theta) - Fg;
    der(x) = v_x;
    der(v_x)*m = (a*u - Fd)*cos(theta);
    v_squared = v_x^2+v_y^2;
    // Inputs
    der(u) = du;
    der(theta) = dtheta;
end Rocket;

optimization OptRocket(finalTime(free=true, min=startTime, initialGuess=280),
                       objective=finalTime, objectiveIntegrand=0.2*du^2 + 0.2*dtheta^2)
    extends Rocket;
constraint
    y(finalTime) = 0;
    x(finalTime) = 3.20e6;
end OptRocket;

optimization OptRocketTime(finalTime(free=true, min=250, max=350, initialGuess=280),
                       objective=finalTime)
    extends Rocket;
constraint
    y(finalTime) = 0;
    x(finalTime) = 3.20e6;
end OptRocketTime;
```